

### 3.4 Instruction Set

The complete MSP430 instruction set consists of 27 core instructions and 24 emulated instructions. The core instructions are instructions that have unique op-codes decoded by the CPU. The emulated instructions are instructions that make code easier to write and read, but do not have op-codes themselves, instead they are replaced automatically by the assembler with an equivalent core instruction. There is no code or performance penalty for using emulated instruction.

There are three core-instruction formats:

- Dual-operand
- Single-operand
- Jump

All single-operand and dual-operand instructions can be byte or word instructions by using .B or .W extensions. Byte instructions are used to access byte data or byte peripherals. Word instructions are used to access word data or word peripherals. If no extension is used, the instruction is a word instruction.

The source and destination of an instruction are defined by the following fields:

src	The source operand defined by As and S-reg
dst	The destination operand defined by Ad and D-reg
As	The addressing bits responsible for the addressing mode used for the source (src)
S-reg	The working register used for the source (src)
Ad	The addressing bits responsible for the addressing mode used for the destination (dst)
D-reg	The working register used for the destination (dst)
B/W	Byte or word operation: 0: word operation 1: byte operation

---


#### **NOTE: Destination Address**

Destination addresses are valid anywhere in the memory map. However, when using an instruction that modifies the contents of the destination, the user must ensure the destination address is writable. For example, a masked-ROM location would be a valid destination address, but the contents are not modifiable, so the results of the instruction would be lost.

---

### 3.4.1 Double-Operand (Format I) Instructions

Figure 3-9 illustrates the double-operand instruction format.



**Figure 3-9. Double Operand Instruction Format**

Table 3-11 lists and describes the double operand instructions.

**Table 3-11. Double Operand Instructions**

Mnemonic	S-Reg, D-Reg	Operation	Status Bits			
			V	N	Z	C
MOV (.B)	src, dst	src → dst	-	-	-	-
ADD (.B)	src, dst	src + dst → dst	*	*	*	*
ADDC (.B)	src, dst	src + dst + C → dst	*	*	*	*
SUB (.B)	src, dst	dst + .not.src + 1 → dst	*	*	*	*
SUBC (.B)	src, dst	dst + .not.src + C → dst	*	*	*	*
CMP (.B)	src, dst	dst - src	*	*	*	*
DADD (.B)	src, dst	src + dst + C → dst (decimally)	*	*	*	*
BIT (.B)	src, dst	src .and. dst	0	*	*	*
BIC (.B)	src, dst	not.src .and. dst → dst	-	-	-	-
BIS (.B)	src, dst	src .or. dst → dst	-	-	-	-
XOR (.B)	src, dst	src .xor. dst → dst	*	*	*	*
AND (.B)	src, dst	src .and. dst → dst	0	*	*	*

- \* The status bit is affected
- The status bit is not affected
- 0 The status bit is cleared
- 1 The status bit is set

**NOTE: Instructions CMP and SUB**

The instructions CMP and SUB are identical except for the storage of the result. The same is true for the BIT and AND instructions.

### 3.4.2 Single-Operand (Format II) Instructions

Figure 3-10 illustrates the single-operand instruction format.

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
Op-code									B/W	Ad		D/S-Reg			

**Figure 3-10. Single Operand Instruction Format**

Table 3-12 lists and describes the single operand instructions.

**Table 3-12. Single Operand Instructions**


Mnemonic	S-Reg, D-Reg	Operation	Status Bits			
			V	N	Z	C
RRC (.B)	dst	C → MSB → .....LSB → C	*	*	*	*
RRA (.B)	dst	MSB → MSB → .....LSB → C	0	*	*	*
PUSH (.B)	src	SP – 2 → SP, src → @SP	-	-	-	-
SWPB	dst	Swap bytes	-	-	-	-
CALL	dst	SP – 2 → SP, PC+2 → @SP	-	-	-	-
RETI		dst → PC	*	*	*	*
		TOS → SR, SP + 2 → SP				
		TOS → PC, SP + 2 → SP				
SXT	dst	Bit 7 → Bit 8.....Bit 15	0	*	*	*

- \* The status bit is affected
- The status bit is not affected
- 0 The status bit is cleared
- 1 The status bit is set

All addressing modes are possible for the CALL instruction. If the symbolic mode (ADDRESS), the immediate mode (#N), the absolute mode (&EDE) or the indexed mode x(RN) is used, the word that follows contains the address information.

### 3.4.3 Jumps

Figure 3-11 shows the conditional-jump instruction format.



**Figure 3-11. Jump Instruction Format**

Table 3-13 lists and describes the jump instructions

**Table 3-13. Jump Instructions**

Mnemonic	S-Reg, D-Reg	Operation
JEQ/JZ	Label	Jump to label if zero bit is set
JNE/JNZ	Label	Jump to label if zero bit is reset
JC	Label	Jump to label if carry bit is set
JNC	Label	Jump to label if carry bit is reset
JN	Label	Jump to label if negative bit is set
JGE	Label	Jump to label if (N .XOR. V) = 0
JL	Label	Jump to label if (N .XOR. V) = 1
JMP	Label	Jump to label unconditionally

Conditional jumps support program branching relative to the PC and do not affect the status bits. The possible jump range is from –511 to +512 words relative to the PC value at the jump instruction. The 10-bit program-counter offset is treated as a signed 10-bit value that is doubled and added to the program counter:

$$PC_{\text{new}} = PC_{\text{old}} + 2 + PC_{\text{offset}} \times 2$$

### 3.4.4 Instruction Cycles and Lengths

The number of CPU clock cycles required for an instruction depends on the instruction format and the addressing modes used - not the instruction itself. The number of clock cycles refers to the MCLK.

#### 3.4.4.1 Interrupt and Reset Cycles

Table 3-14 lists the CPU cycles for interrupt overhead and reset.

**Table 3-14. Interrupt and Reset Cycles**

Action	No. of Cycles	Length of Instruction
Return from interrupt (RETI)	5	1
Interrupt accepted	6	-
WDT reset	4	-
Reset (RST/NMI)	4	-

#### 3.4.4.2 Format-II (Single Operand) Instruction Cycles and Lengths

Table 3-15 lists the length and CPU cycles for all addressing modes of format-II instructions.

**Table 3-15. Format-II Instruction Cycles and Lengths**

Addressing Mode	No. of Cycles			Length of Instruction	Example
	RRA, RRC SWPB, SXT	PUSH	CALL		
Rn	1	3	4	1	SWPB R5
@Rn	3	4	4	1	RRC @R9
@Rn+	3	5	5	1	SWPB @R10+
#N	(See note)	4	5	2	CALL #0F000h
X(Rn)	4	5	5	2	CALL 2 (R7)
EDE	4	5	5	2	PUSH EDE
&EDE	4	5	5	2	SXT &EDE

**NOTE: Instruction Format II Immediate Mode**

Do not use instruction RRA, RRC, SWPB, and SXT with the immediate mode in the destination field. Use of these in the immediate mode results in an unpredictable program operation.

#### 3.4.4.3 Format-III (Jump) Instruction Cycles and Lengths

All jump instructions require one code word, and take two CPU cycles to execute, regardless of whether the jump is taken or not.

### 3.4.4.4 Format-I (Double Operand) Instruction Cycles and Lengths

Table 3-16 lists the length and CPU cycles for all addressing modes of format-I instructions.

**Table 3-16. Format 1 Instruction Cycles and Lengths**

Addressing Mode		No. of Cycles	Length of Instruction		Example
Src	Dst				
Rn	Rm	1	1	MOV	R5, R8
	PC	2	1	BR	R9
	x(Rm)	4	2	ADD	R5, 4 (R6)
	EDE	4	2	XOR	R8, EDE
	&EDE	4	2	MOV	R5, &EDE
@Rn	Rm	2	1	AND	@R4, R5
	PC	2	1	BR	@R8
	x(Rm)	5	2	XOR	@R5, 8 (R6)
	EDE	5	2	MOV	@R5, EDE
	&EDE	5	2	XOR	@R5, &EDE
@Rn+	Rm	2	1	ADD	@R5+, R6
	PC	3	1	BR	@R9+
	x(Rm)	5	2	XOR	@R5, 8 (R6)
	EDE	5	2	MOV	@R9+, EDE
	&EDE	5	2	MOV	@R9+, &EDE
#N	Rm	2	2	MOV	#20, R9
	PC	3	2	BR	#2AEh
	x(Rm)	5	3	MOV	#0300h, 0 (SP)
	EDE	5	3	ADD	#33, EDE
	&EDE	5	3	ADD	#33, &EDE
x(Rn)	Rm	3	2	MOV	2 (R5), R7
	PC	3	2	BR	2 (R6)
	TONI	6	3	MOV	4 (R7), TONI
	x(Rm)	6	3	ADD	4 (R4), 6 (R9)
	&TONI	6	3	MOV	2 (R4), &TONI
EDE	Rm	3	2	AND	EDE, R6
	PC	3	2	BR	EDE
	TONI	6	3	CMP	EDE, TONI
	x(Rm)	6	3	MOV	EDE, 0 (SP)
	&TONI	6	3	MOV	EDE, &TONI
&EDE	Rm	3	2	MOV	&EDE, R8
	PC	3	2	BRA	&EDE
	TONI	6	3	MOV	&EDE, TONI
	x(Rm)	6	3	MOV	&EDE, 0 (SP)
	&TONI	6	3	MOV	&EDE, &TONI

### 3.4.5 Instruction Set Description

The instruction map is shown in Figure 3-12 and the complete instruction set is summarized in Table 3-17.

	000	040	080	0C0	100	140	180	1C0	200	240	280	2C0	300	340	380	3C0
0xxx																
4xxx																
8xxx																
Cxxx																
1xxx	RRC	RRC.B	SWPB		RRA	RRA.B	SXT		PUSH	PUSH.B	CALL		RETI			
14xx																
18xx																
1Cxx																
20xx	JNE/JNZ															
24xx	JEQ/JZ															
28xx	JNC															
2Cxx	JC															
30xx	JN															
34xx	JGE															
38xx	JL															
3Cxx	JMP															
4xxx	MOV, MOV.B															
5xxx	ADD, ADD.B															
6xxx	ADDC, ADDC.B															
7xxx	SUBC, SUBC.B															
8xxx	SUB, SUB.B															
9xxx	CMP, CMP.B															
Axxx	DADD, DADD.B															
Bxxx	BIT, BIT.B															
Cxxx	BIC, BIC.B															
Dxxx	BIS, BIS.B															
Exxx	XOR, XOR.B															
Fxxx	AND, AND.B															

Figure 3-12. Core Instruction Map

Table 3-17. MSP430 Instruction Set

Mnemonic		Description		V	N	Z	C
ADC (.B) <sup>(1)</sup>	dst	Add C to destination	dst + C → dst	*	*	*	*
ADD (.B)	src, dst	Add source to destination	src + dst → dst	*	*	*	*
ADDC (.B)	src, dst	Add source and C to destination	src + dst + C → dst	*	*	*	*
AND (.B)	src, dst	AND source and destination	src .and. dst → dst	0	*	*	*
BIC (.B)	src, dst	Clear bits in destination	not.src .and. dst → dst	-	-	-	-
BIS (.B)	src, dst	Set bits in destination	src .or. dst → dst	-	-	-	-
BIT (.B)	src, dst	Test bits in destination	src .and. dst	0	*	*	*
BR <sup>(1)</sup>	dst	Branch to destination	dst → PC	-	-	-	-
CALL	dst	Call destination	PC+2 → stack, dst → PC	-	-	-	-
CLR (.B) <sup>(1)</sup>	dst	Clear destination	0 → dst	-	-	-	-
CLRC <sup>(1)</sup>		Clear C	0 → C	-	-	-	0
CLR N <sup>(1)</sup>		Clear N	0 → N	-	0	-	-
CLR Z <sup>(1)</sup>		Clear Z	0 → Z	-	-	0	-
CMP (.B)	src, dst	Compare source and destination	dst - src	*	*	*	*
DADC (.B) <sup>(1)</sup>	dst	Add C decimally to destination	dst + C → dst (decimally)	*	*	*	*
DADD (.B)	src, dst	Add source and C decimally to dst	src + dst + C → dst (decimally)	*	*	*	*
DEC (.B) <sup>(1)</sup>	dst	Decrement destination	dst - 1 → dst	*	*	*	*

<sup>(1)</sup> Emulated Instruction

**Table 3-17. MSP430 Instruction Set (continued)**

Mnemonic		Description		V	N	Z	C
DECD (.B) <sup>(1)</sup>	dst	Double-decrement destination	dst - 2 → dst	*	*	*	*
DINT <sup>(1)</sup>		Disable interrupts	0 → GIE	-	-	-	-
EINT <sup>(1)</sup>		Enable interrupts	1 → GIE	-	-	-	-
INC (.B) <sup>(1)</sup>	dst	Increment destination	dst +1 → dst	*	*	*	*
INCD (.B) <sup>(1)</sup>	dst	Double-increment destination	dst+2 → dst	*	*	*	*
INV (.B) <sup>(1)</sup>	dst	Invert destination	.not.dst → dst	*	*	*	*
JC/JHS	label	Jump if C set/Jump if higher or same		-	-	-	-
JEQ/JZ	label	Jump if equal/Jump if Z set		-	-	-	-
JGE	label	Jump if greater or equal		-	-	-	-
JL	label	Jump if less		-	-	-	-
JMP	label	Jump	PC + 2 × offset → PC	-	-	-	-
JN	label	Jump if N set		-	-	-	-
JNC/JLO	label	Jump if C not set/Jump if lower		-	-	-	-
JNE/JNZ	label	Jump if not equal/Jump if Z not set		-	-	-	-
MOV (.B)	src, dst	Move source to destination	src → dst	-	-	-	-
NOP <sup>(2)</sup>		No operation		-	-	-	-
POP (.B) <sup>(2)</sup>	dst	Pop item from stack to destination	@SP → dst, SP+2 → SP	-	-	-	-
PUSH (.B)	src	Push source onto stack	SP - 2 → SP, src → @SP	-	-	-	-
RET <sup>(2)</sup>		Return from subroutine	@SP → PC, SP + 2 → SP	-	-	-	-
RETI		Return from interrupt		*	*	*	*
RLA (.B) <sup>(2)</sup>	dst	Rotate left arithmetically		*	*	*	*
RLC (.B) <sup>(2)</sup>	dst	Rotate left through C		*	*	*	*
RRA (.B)	dst	Rotate right arithmetically		0	*	*	*
RRC (.B)	dst	Rotate right through C		*	*	*	*
SBC (.B) <sup>(2)</sup>	dst	Subtract not(C) from destination	dst + 0FFFFh + C → dst	*	*	*	*
SETC <sup>(2)</sup>		Set C	1 → C	-	-	-	1
SETN <sup>(2)</sup>		Set N	1 → N	-	1	-	-
SETZ <sup>(2)</sup>		Set Z	1 → Z	-	-	1	-
SUB (.B)	src, dst	Subtract source from destination	dst + .not.src + 1 → dst	*	*	*	*
SUBC (.B)	src, dst	Subtract source and not(C) from dst	dst + .not.src + C → dst	*	*	*	*
SWPB	dst	Swap bytes		-	-	-	-
SXT	dst	Extend sign		0	*	*	*
TST (.B) <sup>(2)</sup>	dst	Test destination	dst + 0FFFFh + 1	0	*	*	1
XOR (.B)	src, dst	Exclusive OR source and destination	src .xor. dst → dst	*	*	*	*

<sup>(2)</sup> Emulated Instruction

### 3.4.6 Instruction Set Details

#### 3.4.6.1 ADC

<b>*ADC[.W]</b>	Add carry to destination
<b>*ADC.B</b>	Add carry to destination
<b>Syntax</b>	ADC dst or ADC.W dst ADC.B dst
<b>Operation</b>	dst + C → dst
<b>Emulation</b>	ADDC #0, dst ADDC.B #0, dst
<b>Description</b>	The carry bit (C) is added to the destination operand. The previous contents of the destination are lost.
<b>Status Bit</b>	N: Set if result is negative, reset if positive Z: Set if result is zero, reset otherwise C: Set if dst was incremented from 0FFFFh to 0000, reset otherwise Set if dst was incremented from 0FFh to 00, reset otherwise V: Set if an arithmetic overflow occurs, otherwise reset
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	The 16-bit counter pointed to by R13 is added to a 32-bit counter pointed to by R12. ADD @R13, 0 (R12) ; Add LSDs ADC 2 (R12) ; Add carry to MSD
<b>Example</b>	The 8-bit counter pointed to by R13 is added to a 16-bit counter pointed to by R12. ADD.B @R13, 0 (R12) ; Add LSDs ADC.B 1 (R12) ; Add carry to MSD

### 3.4.6.2 ADD

<b>ADD[.W]</b>	Add source to destination
<b>ADD.B</b>	Add source to destination
<b>Syntax</b>	<pre>ADD src,dst or ADD.W src,dst ADD.B src,dst</pre>
<b>Operation</b>	$\text{src} + \text{dst} \rightarrow \text{dst}$
<b>Description</b>	The source operand is added to the destination operand. The source operand is not affected. The previous contents of the destination are lost.
<b>Status Bits</b>	<p>N: Set if result is negative, reset if positive</p> <p>Z: Set if result is zero, reset otherwise</p> <p>C: Set if there is a carry from the result, cleared if not</p> <p>V: Set if an arithmetic overflow occurs, otherwise reset</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>R5 is increased by 10. The jump to TONI is performed on a carry.</p> <pre>ADD    #10,R5 JC     TONI    ; Carry occurred .....    ; No carry</pre>
<b>Example</b>	<p>R5 is increased by 10. The jump to TONI is performed on a carry.</p> <pre>ADD.B  #10,R5    ; Add 10 to Lowbyte of R5 JC     TONI    ; Carry occurred, if (R5) ≥ 246 [0Ah+0F6h] .....    ; No carry</pre>

### 3.4.6.3 ADDC

<b>ADDC[.W]</b>	Add source and carry to destination
<b>ADDC.B</b>	Add source and carry to destination
<b>Syntax</b>	<pre>ADDC src,dst or ADDC.W src,dst ADDC.B src,dst</pre>
<b>Operation</b>	$\text{src} + \text{dst} + \text{C} \rightarrow \text{dst}$
<b>Description</b>	The source operand and the carry bit (C) are added to the destination operand. The source operand is not affected. The previous contents of the destination are lost.
<b>Status Bits</b>	<p>N: Set if result is negative, reset if positive</p> <p>Z: Set if result is zero, reset otherwise</p> <p>C: Set if there is a carry from the MSB of the result, reset otherwise</p> <p>V: Set if an arithmetic overflow occurs, otherwise reset</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The 32-bit counter pointed to by R13 is added to a 32-bit counter, eleven words (20/2 + 2/2) above the pointer in R13.</p> <pre>ADD      @R13+,20(R13)    ; ADD LSDs with no carry in ADDC     @R13+,20(R13)    ; ADD MSDs with carry ...                               ; resulting from the LSDs</pre>
<b>Example</b>	<p>The 24-bit counter pointed to by R13 is added to a 24-bit counter, eleven words above the pointer in R13.</p> <pre>ADD.B    @R13+,10(R13)    ; ADD LSDs with no carry in ADDC.B   @R13+,10(R13)    ; ADD medium Bits with carry ADDC.B   @R13+,10(R13)    ; ADD MSDs with carry ...                               ; resulting from the LSDs</pre>

### 3.4.6.4 AND

<b>AND[.W]</b>	Source AND destination
<b>AND.B</b>	Source AND destination
<b>Syntax</b>	<pre>AND src,dst or AND.W src,dst AND.B src,dst</pre>
<b>Operation</b>	<code>src .AND. dst → dst</code>
<b>Description</b>	The source operand and the destination operand are logically ANDed. The result is placed into the destination.
<b>Status Bits</b>	<p>N: Set if result MSB is set, reset if not set</p> <p>Z: Set if result is zero, reset otherwise</p> <p>C: Set if result is not zero, reset otherwise ( = .NOT. Zero)</p> <p>V: Reset</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The bits set in R5 are used as a mask (#0AA55h) for the word addressed by TOM. If the result is zero, a branch is taken to label TONI.</p> <pre>MOV      #0AA55h,R5      ; Load mask into register R5 AND      R5,TOM           ; mask word addressed by TOM with R5 JZ       TONI             ; .....                   ; Result is not zero ; ; ;   or ; ; AND      #0AA55h,TOM JZ       TONI</pre>
<b>Example</b>	<p>The bits of mask #0A5h are logically ANDed with the low byte TOM. If the result is zero, a branch is taken to label TONI.</p> <pre>AND.B    #0A5h,TOM       ; mask Lowbyte TOM with 0A5h JZ       TONI             ; .....                   ; Result is not zero</pre>

### 3.4.6.5 BIC

<b>BIC[.W]</b>	Clear bits in destination
<b>BIC.B</b>	Clear bits in destination
<b>Syntax</b>	<pre>BIC src,dst or BIC.W src,dst BIC.B src,dst</pre>
<b>Operation</b>	.NOT.src .AND. dst → dst
<b>Description</b>	The inverted source operand and the destination operand are logically ANDed. The result is placed into the destination. The source operand is not affected.
<b>Status Bits</b>	Status bits are not affected.
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The six MSBs of the RAM word LEO are cleared.</p> <pre>BIC    #0FC00h,LEO ; Clear 6 MSBs in MEM(LEO)</pre>
<b>Example</b>	<p>The five MSBs of the RAM byte LEO are cleared.</p> <pre>BIC.B  #0F8h,LEO   ; Clear 5 MSBs in Ram location LEO</pre>

### 3.4.6.6 BIS

<b>BIS[W]</b>	Set bits in destination
<b>BIS.B</b>	Set bits in destination
<b>Syntax</b>	<pre>BIS src,dst or BIS.W src,dst BIS.B src,dst</pre>
<b>Operation</b>	src .OR. dst → dst
<b>Description</b>	The source operand and the destination operand are logically ORed. The result is placed into the destination. The source operand is not affected.
<b>Status Bits</b>	Status bits are not affected.
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The six LSBs of the RAM word TOM are set.</p> <pre>BIS    #003Fh,TOM    ; set the six LSBs in RAM location TOM</pre>
<b>Example</b>	<p>The three MSBs of RAM byte TOM are set.</p> <pre>BIS.B  #0E0h,TOM     ; set the 3 MSBs in RAM location TOM</pre>

### 3.4.6.7 BIT

<b>BIT[.W]</b>	Test bits in destination
<b>BIT.B</b>	Test bits in destination
<b>Syntax</b>	BIT src,dst or BIT.W src,dst
<b>Operation</b>	src .AND. dst
<b>Description</b>	The source and destination operands are logically ANDed. The result affects only the status bits. The source and destination operands are not affected.
<b>Status Bits</b>	<p>N: Set if MSB of result is set, reset otherwise</p> <p>Z: Set if result is zero, reset otherwise</p> <p>C: Set if result is not zero, reset otherwise (.NOT. Zero)</p> <p>V: Reset</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>If bit 9 of R8 is set, a branch is taken to label TOM.</p> <pre> BIT    #0200h,R8    ; bit 9 of R8 set? JNZ    TOM           ; Yes, branch to TOM ...           ; No, proceed </pre>
<b>Example</b>	<p>If bit 3 of R8 is set, a branch is taken to label TOM.</p> <pre> BIT.B   #8,R8 JC      TOM </pre>
<b>Example</b>	<p>A serial communication receive bit (RCV) is tested. Because the carry bit is equal to the state of the tested bit while using the BIT instruction to test a single bit, the carry bit is used by the subsequent instruction; the read information is shifted into register RECBUF.</p> <pre> ; ; Serial communication with LSB is shifted first: ;      xxxx  xxxx  xxxx  xxxx BIT.B   #RCV,RCCTL  ; Bit info into carry RRC     RECBUF      ; Carry -&gt; MSB of RECBUF ;      cxxx  cxxx .....   ; repeat previous two instructions .....   ; 8 times ;      cccc  cccc ;      ^      ^ ;      MSB   LSB ; Serial communication with MSB shifted first: BIT.B   #RCV,RCCTL  ; Bit info into carry RLC.B   RECBUF      ; Carry -&gt; LSB of RECBUF ;      xxxx  xxxc .....   ; repeat previous two instructions .....   ; 8 times ;      cccc  cccc ;        ;      MSB   LSB </pre>

### 3.4.6.8 BR, BRANCH

<b>*BR, BRANCH</b>	Branch to ..... destination
<b>Syntax</b>	BR dst
<b>Operation</b>	dst → PC
<b>Emulation</b>	MOV dst, PC
<b>Description</b>	An unconditional branch is taken to an address anywhere in the 64K address space. All source addressing modes can be used. The branch instruction is a word instruction.
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>Examples for all addressing modes are given.</p> <pre> BR #EXEC ; Branch to label EXEC or direct branch (e.g. #0A4h)           ; Core instruction MOV @PC+,PC BR EXEC  ; Branch to the address contained in EXEC           ; Core instruction MOV X(PC),PC           ; Indirect address BR &amp;EXEC ; Branch to the address contained in absolute           ; address EXEC           ; Core instruction MOV X(0),PC           ; Indirect address BR R5    ; Branch to the address contained in R5           ; Core instruction MOV R5,PC           ; Indirect R5 BR @R5   ; Branch to the address contained in the word           ; pointed to by R5.           ; Core instruction MOV @R5+,PC           ; Indirect, indirect R5 BR @R5+  ; Branch to the address contained in the word pointed           ; to by R5 and increment pointer in R5 afterwards.           ; The next time--S/W flow uses R5 pointer--it can           ; alter program execution due to access to           ; next address in a table pointed to by R5           ; Core instruction MOV @R5,PC           ; Indirect, indirect R5 with autoincrement BR X(R5) ; Branch to the address contained in the address           ; pointed to by R5 + X (e.g. table with address           ; starting at X). X can be an address or a label           ; Core instruction MOV X(R5),PC           ; Indirect, indirect R5 + X </pre>

### 3.4.6.9 CALL

<b>CALL</b>	Subroutine
<b>Syntax</b>	CALL dst
<b>Operation</b>	dst → tmp      dst is evaluated and stored SP - 2 → SP PC → @SP      PC updated to TOS tmp → PC      dst saved to PC
<b>Description</b>	A subroutine call is made to an address anywhere in the 64K address space. All addressing modes can be used. The return address (the address of the following instruction) is stored on the stack. The call instruction is a word instruction.
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	Examples for all addressing modes are given. <pre> CALL #EXEC ; Call on label EXEC or immediate address (e.g. #0A4h)             ; SP-2 -&gt; SP, PC+2 -&gt; @SP, @PC+ -&gt; PC CALL EXEC  ; Call on the address contained in EXEC             ; SP-2 -&gt; SP, PC+2 -&gt; @SP, X(PC) -&gt; PC             ; Indirect address CALL &amp;EXEC ; Call on the address contained in absolute address             ; EXEC             ; SP-2 -&gt; SP, PC+2 -&gt; @SP, X(0) -&gt; PC             ; Indirect address CALL R5    ; Call on the address contained in R5             ; SP-2 -&gt; SP, PC+2 -&gt; @SP, R5 -&gt; PC             ; Indirect R5 CALL @R5   ; Call on the address contained in the word             ; pointed to by R5             ; SP-2 -&gt; SP, PC+2 -&gt; @SP, @R5 -&gt; PC             ; Indirect, indirect R5 CALL @R5+  ; Call on the address contained in the word             ; pointed to by R5 and increment pointer in R5.             ; The next time S/W flow uses R5 pointer             ; it can alter the program execution due to             ; access to next address in a table pointed to by R5             ; SP-2 -&gt; SP, PC+2 -&gt; @SP, @R5 -&gt; PC             ; Indirect, indirect R5 with autoincrement CALL X(R5) ; Call on the address contained in the address pointed             ; to by R5 + X (e.g. table with address starting at X)             ; X can be an address or a label             ; SP-2 -&gt; SP, PC+2 -&gt; @SP, X(R5) -&gt; PC             ; Indirect, indirect R5 + X           </pre>

### 3.4.6.10 CLR

<b>*CLR[W]</b>	Clear destination
<b>*CLR.B</b>	Clear destination
<b>Syntax</b>	<pre>CLR dst or CLR.W dst CLR.B dst</pre>
<b>Operation</b>	$0 \rightarrow \text{dst}$
<b>Emulation</b>	<pre>MOV #0, dst MOV.B #0, dst</pre>
<b>Description</b>	The destination operand is cleared.
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>RAM word TONI is cleared.</p> <pre>CLR    TONI    ; 0 -&gt; TONI</pre>
<b>Example</b>	<p>Register R5 is cleared.</p> <pre>CLR    R5</pre>
<b>Example</b>	<p>RAM byte TONI is cleared.</p> <pre>CLR.B  TONI    ; 0 -&gt; TONI</pre>

### 3.4.6.11 CLRC

<b>*CLRC</b>	Clear carry bit
<b>Syntax</b>	CLRC
<b>Operation</b>	$0 \rightarrow C$
<b>Emulation</b>	BIC #1,SR
<b>Description</b>	The carry bit (C) is cleared. The clear carry instruction is a word instruction.
<b>Status Bits</b>	N: Not affected Z: Not affected C: Cleared V: Not affected
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The 16-bit decimal counter pointed to by R13 is added to a 32-bit counter pointed to by R12.</p> <pre> CLRC                ; C=0: defines start DADD    @R13,0(R12) ; add 16-bit counter to low word of 32-bit counter DADC    2(R12)      ; add carry to high word of 32-bit counter </pre>

### 3.4.6.12 CLRN

<b>*CLRN</b>	Clear negative bit
<b>Syntax</b>	CLRN
<b>Operation</b>	$0 \rightarrow N$ or (.NOT.src .AND. dst $\rightarrow$ dst)
<b>Emulation</b>	BIC #4, SR
<b>Description</b>	The constant 04h is inverted (0FFFBh) and is logically ANDed with the destination operand. The result is placed into the destination. The clear negative bit instruction is a word instruction.
<b>Status Bits</b>	N: Reset to 0 Z: Not affected C: Not affected V: Not affected
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The Negative bit in the status register is cleared. This avoids special treatment with negative numbers of the subroutine called.</p> <pre>         CLRN         CALL SUBR         .....         ..... SUBR    JN      SUBRET    ; If input is negative: do nothing and return         .....         .....         ..... SUBRET  RET </pre>

### 3.4.6.13 CLRZ

<b>*CLRZ</b>	Clear zero bit
<b>Syntax</b>	CLRZ
<b>Operation</b>	$0 \rightarrow Z$ or (.NOT.src .AND. dst $\rightarrow$ dst)
<b>Emulation</b>	BIC #2, SR
<b>Description</b>	The constant 02h is inverted (0FFFDh) and logically ANDed with the destination operand. The result is placed into the destination. The clear zero bit instruction is a word instruction.
<b>Status Bits</b>	N: Not affected Z: Reset to 0 C: Not affected V: Not affected
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	The zero bit in the status register is cleared. CLRZ

### 3.4.6.14 CMP

<b>CMP[.W]</b>	Compare source and destination
<b>CMP.B</b>	Compare source and destination
<b>Syntax</b>	<pre>CMP src,dst or CMP.W src,dst CMP.B src,dst</pre>
<b>Operation</b>	$dst + \text{.NOT.src} + 1$ or $(dst - src)$
<b>Description</b>	The source operand is subtracted from the destination operand. This is accomplished by adding the 1s complement of the source operand plus 1. The two operands are not affected and the result is not stored; only the status bits are affected.
<b>Status Bits</b>	N: Set if result is negative, reset if positive ( $src \geq dst$ ) Z: Set if result is zero, reset otherwise ( $src = dst$ ) C: Set if there is a carry from the MSB of the result, reset otherwise V: Set if an arithmetic overflow occurs, otherwise reset
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>R5 and R6 are compared. If they are equal, the program continues at the label EQUAL.</p> <pre>CMP    R5,R6    ; R5 = R6? JEQ    EQUAL    ; YES, JUMP</pre>
<b>Example</b>	<p>Two RAM blocks are compared. If they are not equal, the program branches to the label ERROR.</p> <pre>MOV    #NUM,R5      ; number of words to be compared MOV    #BLOCK1,R6   ; BLOCK1 start address in R6 MOV    #BLOCK2,R7   ; BLOCK2 start address in R7 L\$1    CMP    @R6+,0(R7) ; Are Words equal? R6 increments JNZ    ERROR        ; No, branch to ERROR INCD   R7           ; Increment R7 pointer DEC    R5           ; Are all words compared? JNZ    L\$1          ; No, another compare</pre>
<b>Example</b>	<p>The RAM bytes addressed by EDE and TONI are compared. If they are equal, the program continues at the label EQUAL.</p> <pre>CMP.B  EDE,TONI    ; MEM(EDE) = MEM(TONI)? JEQ     EQUAL      ; YES, JUMP</pre>

### 3.4.6.15 DADC

<b>*DADC[W]</b>	Add carry decimally to destination
<b>*DADC.B</b>	Add carry decimally to destination
<b>Syntax</b>	DADC dst or DADC.W src,dst DADC.B dst
<b>Operation</b>	dst + C → dst (decimally)
<b>Emulation</b>	DADD #0,dst DADD.B #0,dst
<b>Description</b>	The carry bit (C) is added decimally to the destination.
<b>Status Bits</b>	N: Set if MSB is 1 Z: Set if dst is 0, reset otherwise C: Set if destination increments from 9999 to 0000, reset otherwise Set if destination increments from 99 to 00, reset otherwise V: Undefined
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The four-digit decimal number contained in R5 is added to an eight-digit decimal number pointed to by R8.</p> <pre> CLRC                ; Reset carry                     ; next instruction's start condition is defined DADD    R5,0(R8)    ; Add LSDs + C DADC    2(R8)        ; Add carry to MSD </pre>
<b>Example</b>	<p>The two-digit decimal number contained in R5 is added to a four-digit decimal number pointed to by R8.</p> <pre> CLRC                ; Reset carry                     ; next instruction's start condition is defined DADD.B  R5,0(R8)    ; Add LSDs + C DADC.B  1(R8)        ; Add carry to MSDs </pre>


### 3.4.6.16 DADD

<b>DADD[.W]</b>	Source and carry added decimally to destination
<b>DADD.B</b>	Source and carry added decimally to destination
<b>Syntax</b>	DADD src,dst or DADD.W src,dst DADD.B src,dst
<b>Operation</b>	$\text{src} + \text{dst} + \text{C} \rightarrow \text{dst}$ (decimally)
<b>Description</b>	The source operand and the destination operand are treated as four binary coded decimals (BCD) with positive signs. The source operand and the carry bit (C) are added decimally to the destination operand. The source operand is not affected. The previous contents of the destination are lost. The result is not defined for non-BCD numbers.
<b>Status Bits</b>	N: Set if the MSB is 1, reset otherwise Z: Set if result is zero, reset otherwise C: Set if the result is greater than 9999 Set if the result is greater than 99 V: Undefined
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The eight-digit BCD number contained in R5 and R6 is added decimally to an eight-digit BCD number contained in R3 and R4 (R6 and R4 contain the MSDs).</p> <pre> CLRC                ; clear carry DADD    R5,R3        ; add LSDs DADD    R6,R4        ; add MSDs with carry JC      OVERFLOW     ; If carry occurs go to error handling routine </pre>
<b>Example</b>	<p>The two-digit decimal counter in the RAM byte CNT is incremented by one.</p> <pre> CLRC                ; clear carry DADD.B  #1,CNT </pre> <p>or</p> <pre> SETC DADD.B  #0,CNT      ; equivalent to DADC.B CNT </pre>

### 3.4.6.17 DEC

<b>*DEC[W]</b>	Decrement destination
<b>*DEC.B</b>	Decrement destination
<b>Syntax</b>	<pre>DEC dst or DEC.W dst DEC.B dst</pre>
<b>Operation</b>	$\text{dst} - 1 \rightarrow \text{dst}$
<b>Emulation</b>	<pre>SUB #1, dst SUB.B #1, dst</pre>
<b>Description</b>	The destination operand is decremented by one. The original contents are lost.
<b>Status Bits</b>	<p>N: Set if result is negative, reset if positive</p> <p>Z: Set if dst contained 1, reset otherwise</p> <p>C: Reset if dst contained 0, set otherwise</p> <p>V: Set if an arithmetic overflow occurs, otherwise reset.</p> <p>Set if initial value of destination was 08000h, otherwise reset.</p> <p>Set if initial value of destination was 080h, otherwise reset.</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>R10 is decremented by 1.</p> <pre>DEC    R10    ; Decrement R10</pre> <p>; Move a block of 255 bytes from memory location starting with EDE to memory location starting with ; TONI. Tables should not overlap: start of destination address TONI must not be within the range EDE ; to EDE+0FEh</p> <pre>MOV    #EDE, R6 MOV    #255, R10 L\$1 MOV.B @R6+, TONI-EDE-1 (R6) DEC    R10 JNZ    L\$1</pre>

Do not transfer tables using the routine above with the overlap shown in [Figure 3-13](#).



**Figure 3-13. Decrement Overlap**

### 3.4.6.18 DECD

<b>*DECD[W]</b>	Double-decrement destination
<b>*DECD.B</b>	Double-decrement destination
<b>Syntax</b>	DECD dst or DECD.W dst DECD.B dst
<b>Operation</b>	dst - 2 → dst
<b>Emulation</b>	SUB #2, dst
<b>Emulation</b>	SUB.B #2, dst
<b>Description</b>	The destination operand is decremented by two. The original contents are lost.
<b>Status Bits</b>	N: Set if result is negative, reset if positive Z: Set if dst contained 2, reset otherwise C: Reset if dst contained 0 or 1, set otherwise V: Set if an arithmetic overflow occurs, otherwise reset. Set if initial value of destination was 08001 or 08000h, otherwise reset. Set if initial value of destination was 081 or 080h, otherwise reset.
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	R10 is decremented by 2.  <pre> DECD    R10          ; Decrement R10 by two ; Move a block of 255 words from memory location starting with EDE to ; memory location starting with TONI ; Tables should not overlap: start of destination address TONI must not be ; within the range EDE to EDE+0FEh  MOV     #EDE, R6 MOV     #510, R10 L\$1 MOV   @R6+, TONI-EDE-2 (R6) DECD    R10 JNZ     L\$1 </pre>
<b>Example</b>	Memory at location LEO is decremented by two. <pre> DECD.B   LEO          ; Decrement MEM(LEO) </pre> Decrement status byte STATUS by two. <pre> DECD.B   STATUS </pre>

### 3.4.6.19 DINT

<b>*DINT</b>	Disable (general) interrupts
<b>Syntax</b>	DINT
<b>Operation</b>	<p>0 → GIE</p> <p>or</p> <p>(0FFF7h .AND. SR → SR / .NOT.src .AND. dst → dst)</p>
<b>Emulation</b>	BIC #8,SR
<b>Description</b>	<p>All interrupts are disabled.</p> <p>The constant 08h is inverted and logically ANDed with the status register (SR). The result is placed into the SR.</p>
<b>Status Bits</b>	Status bits are not affected.
<b>Mode Bits</b>	GIE is reset. OSCOFF and CPUOFF are not affected.
<b>Example</b>	<p>The general interrupt enable (GIE) bit in the status register is cleared to allow a nondisrupted move of a 32-bit counter. This ensures that the counter is not modified during the move by any interrupt.</p> <pre> DINT          ; All interrupt events using the GIE bit are disabled NOP MOV  COUNTHI,R5 ; Copy counter MOV  COUNTLO,R6 EINT          ; All interrupt events using the GIE bit are enabled </pre>

#### **NOTE: Disable Interrupt**

If any code sequence needs to be protected from interruption, the DINT should be executed at least one instruction before the beginning of the uninterruptible sequence, or should be followed by a NOP instruction.

### 3.4.6.20 EINT

<b>*EINT</b>	Enable (general) interrupts
<b>Syntax</b>	EINT
<b>Operation</b>	1 → GIE or (0008h .OR. SR → SR / .src .OR. dst → dst)
<b>Emulation</b>	BIS #8,SR
<b>Description</b>	All interrupts are enabled.  The constant #08h and the status register SR are logically ORed. The result is placed into the SR.
<b>Status Bits</b>	Status bits are not affected.
<b>Mode Bits</b>	GIE is set. OSCOFF and CPUOFF are not affected.
<b>Example</b>	The general interrupt enable (GIE) bit in the status register is set.

```

; Interrupt routine of ports P1.2 to P1.7
; P1IN is the address of the register where all port bits are read. P1IFG is
; the address of the register where all interrupt events are latched.

        PUSH.B    &P1IN
        BIC.B     @SP,&P1IFG    ; Reset only accepted flags
        EINT      ; Preset port 1 interrupt flags stored on stack
                        ; other interrupts are allowed

        BIT       #Mask,@SP
        JEQ       MaskOK       ; Flags are present identically to mask: jump
        .....
MaskOK   BIC       #Mask,@SP
        .....
        INCD      SP           ; Housekeeping: inverse to PUSH instruction
                                ; at the start of interrupt subroutine. Corrects
                                ; the stack pointer.

        RETI

```

---

**NOTE: Enable Interrupt**

The instruction following the enable interrupt instruction (EINT) is always executed, even if an interrupt service request is pending when the interrupts are enable.

---

### 3.4.6.21 INC

<b>*INC[.W]</b>	Increment destination
<b>*INC.B</b>	Increment destination
<b>Syntax</b>	<pre>INC dst or INC.W dst INC.B dst</pre>
<b>Operation</b>	$dst + 1 \rightarrow dst$
<b>Emulation</b>	<code>ADD #1, dst</code>
<b>Description</b>	The destination operand is incremented by one. The original contents are lost.
<b>Status Bits</b>	<p>N: Set if result is negative, reset if positive</p> <p>Z: Set if dst contained 0FFFFh, reset otherwise Set if dst contained 0FFh, reset otherwise</p> <p>C: Set if dst contained 0FFFFh, reset otherwise Set if dst contained 0FFh, reset otherwise</p> <p>V: Set if dst contained 07FFFh, reset otherwise Set if dst contained 07Fh, reset otherwise</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The status byte, STATUS, of a process is incremented. When it is equal to 11, a branch to OVFL is taken.</p> <pre>INC.B STATUS CMP.B #11, STATUS JEQ OVFL</pre>

### 3.4.6.22 INCD

<b>*INCD[.W]</b>	Double-increment destination
<b>*INCD.B</b>	Double-increment destination
<b>Syntax</b>	INCD dst or INCD.W dst INCD.B dst
<b>Operation</b>	dst + 2 → dst
<b>Emulation</b>	ADD #2, dst ADD.B #2, dst
<b>Example</b>	The destination operand is incremented by two. The original contents are lost.
<b>Status Bits</b>	N: Set if result is negative, reset if positive Z: Set if dst contained 0FFFEh, reset otherwise Set if dst contained 0FEh, reset otherwise C: Set if dst contained 0FFFEh or 0FFFFh, reset otherwise Set if dst contained 0FEh or 0FFh, reset otherwise V: Set if dst contained 07FFEh or 07FFFh, reset otherwise Set if dst contained 07Eh or 07Fh, reset otherwise
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	The item on the top of the stack (TOS) is removed without using a register. <pre> PUSH    R5      ; R5 is the result of a calculation, which is stored                 ; in the system stack INCD     SP      ; Remove TOS by double-increment from stack                 ; Do not use INCD.B, SP is a word-aligned register RET </pre>
<b>Example</b>	The byte on the top of the stack is incremented by two. <pre> INCD.B   0(SP)   ; Byte on TOS is increment by two </pre>

### 3.4.6.23 INV

<b>*INV[.W]</b>	Invert destination
<b>*INV.B</b>	Invert destination
<b>Syntax</b>	<pre>INV dst INV.B dst</pre>
<b>Operation</b>	.NOT.dst → dst
<b>Emulation</b>	<pre>XOR #0FFFFh, dst XOR.B #0FFh, dst</pre>
<b>Description</b>	The destination operand is inverted. The original contents are lost.
<b>Status Bits</b>	<p>N: Set if result is negative, reset if positive</p> <p>Z: Set if dst contained 0FFFFh, reset otherwise</p> <p>Set if dst contained 0FFh, reset otherwise</p> <p>C: Set if result is not zero, reset otherwise ( = .NOT. Zero)</p> <p>Set if result is not zero, reset otherwise ( = .NOT. Zero)</p> <p>V: Set if initial destination operand was negative, otherwise reset</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>Content of R5 is negated (twos complement).</p> <pre>MOV    #00AEh, R5    ;          R5 = 000AEh INV     R5            ; Invert R5,    R5 = 0FF51h INC     R5            ; R5 is now negated, R5 = 0FF52h</pre>
<b>Example</b>	<p>Content of memory byte LEO is negated.</p> <pre>MOV.B   #0AEh, LEO    ;          MEM(LEO) = 0AEh INV.B   LEO           ; Invert LEO,    MEM(LEO) = 051h INC.B   LEO           ; MEM(LEO) is negated, MEM(LEO) = 052h</pre>

### 3.4.6.24 JC, JHS

<b>JC</b>	Jump if carry set
<b>JHS</b>	Jump if higher or same
<b>Syntax</b>	JC label JHS label
<b>Operation</b>	If C = 1: PC + 2 offset → PC If C = 0: execute following instruction
<b>Description</b>	The status register carry bit (C) is tested. If it is set, the 10-bit signed offset contained in the instruction LSBs is added to the program counter. If C is reset, the next instruction following the jump is executed. JC (jump if carry/higher or same) is used for the comparison of unsigned numbers (0 to 65536).
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	The P1IN.1 signal is used to define or control the program flow. <pre> BIT.B    #02h,&amp;P1IN    ; State of signal -&gt; Carry JC       PROGA          ; If carry=1 then execute program routine A .....          ; Carry=0, execute program here </pre>
<b>Example</b>	R5 is compared to 15. If the content is higher or the same, branch to LABEL. <pre> CMP      #15,R5 JHS      LABEL          ; Jump is taken if R5 &gt;= 15 .....          ; Continue here if R5 &lt; 15 </pre>

### 3.4.6.25 JEQ, JZ

<b>JEQ, JZ</b>	Jump if equal, jump if zero
<b>Syntax</b>	<pre>JEQ label JZ label</pre>
<b>Operation</b>	<p>If Z = 1: PC + 2 offset → PC</p> <p>If Z = 0: execute following instruction</p>
<b>Description</b>	The status register zero bit (Z) is tested. If it is set, the 10-bit signed offset contained in the instruction LSBs is added to the program counter. If Z is not set, the instruction following the jump is executed.
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>Jump to address TONI if R7 contains zero.</p> <pre>TST    R7                ; Test R7 JZ     TONI              ; if zero: JUMP</pre>
<b>Example</b>	<p>Jump to address LEO if R6 is equal to the table contents.</p> <pre>CMP    R6,Table(R5)      ; Compare content of R6 with content of                         ; MEM (table address + content of R5) JEQ     LEO              ; Jump if both data are equal .....                  ; No, data are not equal, continue here</pre>
<b>Example</b>	<p>Branch to LABEL if R5 is 0.</p> <pre>TST    R5 JZ     LABEL .....</pre>

### 3.4.6.26 JGE

<b>JGE</b>	Jump if greater or equal
<b>Syntax</b>	JGE label
<b>Operation</b>	<p>If (N .XOR. V) = 0 then jump to label: PC + 2 P offset → PC</p> <p>If (N .XOR. V) = 1 then execute the following instruction</p>
<b>Description</b>	<p>The status register negative bit (N) and overflow bit (V) are tested. If both N and V are set or reset, the 10-bit signed offset contained in the instruction LSBs is added to the program counter. If only one is set, the instruction following the jump is executed.</p> <p>This allows comparison of signed integers.</p>
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>When the content of R6 is greater or equal to the memory pointed to by R7, the program continues at label EDE.</p> <pre> CMP    @R7,R6    ; R6 &gt;= (R7)?, compare on signed numbers JGE    EDE       ; Yes, R6 &gt;= (R7) .....          ; No, proceed ..... ..... </pre>

### 3.4.6.27 JL

<b>JL</b>	Jump if less
<b>Syntax</b>	JL label
<b>Operation</b>	<p>If (N .XOR. V) = 1 then jump to label: PC + 2 offset → PC</p> <p>If (N .XOR. V) = 0 then execute following instruction</p>
<b>Description</b>	<p>The status register negative bit (N) and overflow bit (V) are tested. If only one is set, the 10-bit signed offset contained in the instruction LSBs is added to the program counter. If both N and V are set or reset, the instruction following the jump is executed.</p> <p>This allows comparison of signed integers.</p>
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>When the content of R6 is less than the memory pointed to by R7, the program continues at label EDE.</p> <pre> CMP    @R7,R6    ; R6 &lt; (R7)?,  compare on signed numbers JL     EDE       ; Yes, R6 &lt; (R7) .....          ; No, proceed ..... ..... </pre>

---

**3.4.6.28 JMP**

---

<b>JMP</b>	Jump unconditionally
<b>Syntax</b>	JMP label
<b>Operation</b>	$PC + 2 \times \text{offset} \rightarrow PC$
<b>Description</b>	The 10-bit signed offset contained in the instruction LSBs is added to the program counter.
<b>Status Bits</b>	Status bits are not affected.
<b>Hint</b>	This one-word instruction replaces the BRANCH instruction in the range of –511 to +512 words relative to the current program counter.

### 3.4.6.29 JN

<b>JN</b>	Jump if negative
<b>Syntax</b>	JN label
<b>Operation</b>	if N = 1: PC + 2 × offset → PC if N = 0: execute following instruction
<b>Description</b>	The negative bit (N) of the status register is tested. If it is set, the 10-bit signed offset contained in the instruction LSBs is added to the program counter. If N is reset, the next instruction following the jump is executed.
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>The result of a computation in R5 is to be subtracted from COUNT. If the result is negative, COUNT is to be cleared and the program continues execution in another path.</p> <pre> SUB    R5,COUNT    ; COUNT - R5 -&gt; COUNT JN     L\$1          ; If negative continue with COUNT=0 at PC=L\$1 .....             ; Continue with COUNT&gt;=0 ..... ..... ..... L\$1    CLR    COUNT ..... ..... ..... </pre>

### 3.4.6.30 JNC, JLO

<b>JNC</b>	Jump if carry not set
<b>JLO</b>	Jump if lower
<b>Syntax</b>	<pre>JNC label JLO label</pre>
<b>Operation</b>	<p>if C = 0: PC + 2 offset → PC</p> <p>if C = 1: execute following instruction</p>
<b>Description</b>	<p>The status register carry bit (C) is tested. If it is reset, the 10-bit signed offset contained in the instruction LSBs is added to the program counter. If C is set, the next instruction following the jump is executed. JNC (jump if no carry/lower) is used for the comparison of unsigned numbers (0 to 65536).</p>
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>The result in R6 is added in BUFFER. If an overflow occurs, an error handling routine at address ERROR is used.</p> <pre>       ADD    R6,BUFFER    ; BUFFER + R6 -&gt; BUFFER       JNC    CONT        ; No carry, jump to CONT ERROR  .....          ; Error handler start       .....       .....       ..... CONT   .....          ; Continue with normal program flow       .....       ..... </pre>
<b>Example</b>	<p>Branch to STL2 if byte STATUS contains 1 or 0.</p> <pre>       CMP.B  #2,STATUS       JLO    STL 2       ; STATUS &lt; 2       .....           ; STATUS &gt;= 2, continue here </pre>

### 3.4.6.31 JNE, JNZ

<b>JNE</b>	Jump if not equal
<b>JNZ</b>	Jump if not zero
<b>Syntax</b>	<pre>JNE label JNZ label</pre>
<b>Operation</b>	<p>If Z = 0: PC + 2 a offset → PC</p> <p>If Z = 1: execute following instruction</p>
<b>Description</b>	The status register zero bit (Z) is tested. If it is reset, the 10-bit signed offset contained in the instruction LSBs is added to the program counter. If Z is set, the next instruction following the jump is executed.
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>Jump to address TONI if R7 and R8 have different contents.</p> <pre>CMP    R7,R8    ; COMPARE R7 WITH R8 JNE    TONI     ; if different: jump .....        ; if equal, continue</pre>

### 3.4.6.32 MOV

<b>MOV[.W]</b>	Move source to destination
<b>MOV.B</b>	Move source to destination
<b>Syntax</b>	<pre>MOV src,dst or MOV.W src,dst MOV.B src,dst</pre>
<b>Operation</b>	src → dst
<b>Description</b>	<p>The source operand is moved to the destination.</p> <p>The source operand is not affected. The previous contents of the destination are lost.</p>
<b>Status Bits</b>	Status bits are not affected.
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The contents of table EDE (word data) are copied to table TOM. The length of the tables must be 020h locations.</p> <pre> MOV    #EDE,R10                ; Prepare pointer MOV    #020h,R9                ; Prepare counter Loop   MOV    @R10+,TOM-EDE-2(R10) ; Use pointer in R10 for both tables         DEC    R9                ; Decrement counter         JNZ    Loop              ; Counter not 0, continue copying         .....                  ; Copying completed         .....         .....</pre>
<b>Example</b>	<p>The contents of table EDE (byte data) are copied to table TOM. The length of the tables should be 020h locations</p> <pre> MOV    #EDE,R10                ; Prepare pointer MOV    #020h,R9                ; Prepare counter Loop   MOV.B  @R10+,TOM-EDE-1(R10) ; Use pointer in R10 for         ; both tables         DEC    R9                ; Decrement counter         JNZ    Loop              ; Counter not 0, continue         ; copying         .....                  ; Copying completed         .....         .....</pre>

### 3.4.6.33 NOP

<b>*NOP</b>	No operation
<b>Syntax</b>	NOP
<b>Operation</b>	None
<b>Emulation</b>	MOV #0, R3
<b>Description</b>	No operation is performed. The instruction may be used for the elimination of instructions during the software check or for defined waiting times.
<b>Status Bits</b>	Status bits are not affected.  The NOP instruction is mainly used for two purposes: <ul style="list-style-type: none"> <li>• To fill one, two, or three memory words</li> <li>• To adjust software timing</li> </ul>

---

**NOTE: Emulating No-Operation Instruction**

Other instructions can emulate the NOP function while providing different numbers of instruction cycles and code words. Some examples are:

```
MOV  #0,R3           ; 1 cycle, 1 word
MOV  0(R4),0(R4)     ; 6 cycles, 3 words
MOV  @R4,0(R4)       ; 5 cycles, 2 words
BIC  #0,EDE(R4)      ; 4 cycles, 2 words
JMP  $+2             ; 2 cycles, 1 word
BIC  #0,R5           ; 1 cycle, 1 word
```

However, care should be taken when using these examples to prevent unintended results. For example, if MOV 0(R4), 0(R4) is used and the value in R4 is 120h, then a security violation occurs with the watchdog timer (address 120h), because the security key was not used.

---

### 3.4.6.34 POP

<b>*POP[W]</b>	Pop word from stack to destination
<b>*POP.B</b>	Pop byte from stack to destination
<b>Syntax</b>	<pre>POP dst POP.B dst</pre>
<b>Operation</b>	<pre>@SP → temp SP + 2 → SP temp → dst</pre>
<b>Emulation</b>	<pre>MOV @SP+,dst or MOV.W @SP+,dst MOV.B @SP+,dst</pre>
<b>Description</b>	The stack location pointed to by the stack pointer (TOS) is moved to the destination. The stack pointer is incremented by two afterwards.
<b>Status Bits</b>	Status bits are not affected.
<b>Example</b>	<p>The contents of R7 and the status register are restored from the stack.</p> <pre>POP    R7        ; Restore R7 POP    SR        ; Restore status register</pre>
<b>Example</b>	<p>The contents of RAM byte LEO is restored from the stack.</p> <pre>POP.B  LEO       ; The low byte of the stack is moved to LEO.</pre>
<b>Example</b>	<p>The contents of R7 is restored from the stack.</p> <pre>POP.B  R7        ; The low byte of the stack is moved to R7,                   ; the high byte of R7 is 00h</pre>
<b>Example</b>	<p>The contents of the memory pointed to by R7 and the status register are restored from the stack.</p> <pre>POP.B  0(R7)     ; The low byte of the stack is moved to the                   ; the byte which is pointed to by R7                   ; Example:  R7 = 203h                   ;           Mem(R7) = low byte of system stack                   ; Example:  R7 = 20Ah                   ;           Mem(R7) = low byte of system stack POP     SR        ; Last word on stack moved to the SR</pre>

---

**NOTE: The System Stack Pointer**

The system stack pointer (SP) is always incremented by two, independent of the byte suffix.

---

### 3.4.6.35 PUSH

<b>PUSH[W]</b>	Push word onto stack
<b>PUSH.B</b>	Push byte onto stack
<b>Syntax</b>	<pre>PUSH src or PUSH.W src PUSH.B src</pre>
<b>Operation</b>	<p>SP - 2 → SP</p> <p>src → @SP</p>
<b>Description</b>	The stack pointer is decremented by two, then the source operand is moved to the RAM word addressed by the stack pointer (TOS).
<b>Status Bits</b>	Status bits are not affected.
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The contents of the status register and R8 are saved on the stack.</p> <pre>PUSH    SR      ; save status register PUSH    R8      ; save R8</pre>
<b>Example</b>	<p>The contents of the peripheral TCDAT is saved on the stack.</p> <pre>PUSH.B  &amp;TCDAT ; save data from 8-bit peripheral module,                ; address TCDAT, onto stack</pre>

---

**NOTE: System Stack Pointer**

The System stack pointer (SP) is always decremented by two, independent of the byte suffix.

---

---


**3.4.6.36 RET**

---

<b>*RET</b>	Return from subroutine
<b>Syntax</b>	RET
<b>Operation</b>	@SP → PC SP + 2 → SP
<b>Emulation</b>	MOV @SP+, PC
<b>Description</b>	The return address pushed onto the stack by a CALL instruction is moved to the program counter. The program continues at the code address following the subroutine call.
<b>Status Bits</b>	Status bits are not affected.

### 3.4.6.37 RETI


<b>RETI</b>	Return from interrupt
<b>Syntax</b>	RETI
<b>Operation</b>	<p>TOS → SR</p> <p>SP + 2 → SP</p> <p>TOS → PC</p> <p>SP + 2 → SP</p>
<b>Description</b>	<p>The status register is restored to the value at the beginning of the interrupt service routine by replacing the present SR contents with the TOS contents. The stack pointer (SP) is incremented by two.</p> <p>The program counter is restored to the value at the beginning of interrupt service. This is the consecutive step after the interrupted program flow. Restoration is performed by replacing the present PC contents with the TOS memory contents. The stack pointer (SP) is incremented.</p>
<b>Status Bits</b>	<p>N: Restored from system stack</p> <p>Z: Restored from system stack</p> <p>C: Restored from system stack</p> <p>V: Restored from system stack</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are restored from system stack.
<b>Example</b>	<a href="#">Figure 3-14</a> illustrates the main program interrupt.



**Figure 3-14. Main Program Interrupt**

### 3.4.6.38 RLA

<b>*RLA[W]</b>	Rotate left arithmetically
<b>*RLA.B</b>	Rotate left arithmetically
<b>Syntax</b>	<pre>RLA dst or RLA.W dst RLA.B dst</pre>
<b>Operation</b>	$C \leftarrow MSB \leftarrow MSB-1 \dots LSB+1 \leftarrow LSB \leftarrow 0$
<b>Emulation</b>	<pre>ADD dst, dst ADD.B dst, dst</pre>
<b>Description</b>	<p>The destination operand is shifted left one position as shown in <a href="#">Figure 3-15</a>. The MSB is shifted into the carry bit (C) and the LSB is filled with 0. The RLA instruction acts as a signed multiplication by 2.</p> <p>An overflow occurs if <math>dst \geq 04000h</math> and <math>dst &lt; 0C000h</math> before operation is performed: the result has changed sign.</p>



**Figure 3-15. Destination Operand – Arithmetic Shift Left**

An overflow occurs if  $dst \geq 040h$  and  $dst < 0C0h$  before the operation is performed: the result has changed sign.

<b>Status Bits</b>	<p>N: Set if result is negative, reset if positive</p> <p>Z: Set if result is zero, reset otherwise</p> <p>C: Loaded from the MSB</p> <p>V: Set if an arithmetic overflow occurs:  the initial value is <math>04000h \leq dst &lt; 0C000h</math>; reset otherwise  Set if an arithmetic overflow occurs:  the initial value is <math>040h \leq dst &lt; 0C0h</math>; reset otherwise</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>R7 is multiplied by 2.</p> <pre>RLA    R7    ; Shift left R7 (x 2)</pre>
<b>Example</b>	<p>The low byte of R7 is multiplied by 4.</p> <pre>RLA.B  R7    ; Shift left low byte of R7 (x 2) RLA.B  R7    ; Shift left low byte of R7 (x 4)</pre>

**NOTE: RLA Substitution**

The assembler does not recognize the instruction:

```
RLA @R5+, RLA.B @R5+, or RLA(.B) @R5
```

It must be substituted by:

```
ADD @R5+, -2(R5), ADD.B @R5+, -1(R5), or ADD(.B) @R5
```

### 3.4.6.39 RLC

**\*RLC[.W]** Rotate left through carry


**\*RLC.B** Rotate left through carry

**Syntax**  
RLC dst or RLC.W dst  
RLC.B dst

**Operation**  $C \leftarrow \text{MSB} \leftarrow \text{MSB}-1 \dots \text{LSB}+1 \leftarrow \text{LSB} \leftarrow C$

**Emulation** ADDC dst, dst

**Description** The destination operand is shifted left one position as shown in Figure 3-16. The carry bit (C) is shifted into the LSB and the MSB is shifted into the carry bit (C).



**Figure 3-16. Destination Operand - Carry Left Shift**

**Status Bits**

- N: Set if result is negative, reset if positive
- Z: Set if result is zero, reset otherwise
- C: Loaded from the MSB
- V: Set if an arithmetic overflow occurs
  - the initial value is  $04000h \leq \text{dst} < 0C000h$ ; reset otherwise
  - Set if an arithmetic overflow occurs:
  - the initial value is  $040h \leq \text{dst} < 0C0h$ ; reset otherwise

**Mode Bits** OSCOFF, CPUOFF, and GIE are not affected.

**Example** R5 is shifted left one position.  
RLC R5 ; (R5 x 2) + C -> R5

**Example** The input P1IN.1 information is shifted into the LSB of R5.  
BIT.B #2,&P1IN ; Information -> Carry  
RLC R5 ; Carry=P0in.1 -> LSB of R5

**Example** The MEM(LEO) content is shifted left one position.  
RLC.B LEO ; Mem(LEO) x 2 + C -> Mem(LEO)

**NOTE: RLC and RLC.B Substitution**

The assembler does not recognize the instruction:

RLC @R5+, RLC @R5, or RLC(.B) @R5

It must be substituted by:


ADDC @R5+, -2(R5), ADDC.B @R5+, -1(R5), or ADDC(.B) @R5

### 3.4.6.40 RRA

<b>RRA[.W]</b>	Rotate right arithmetically
<b>RRA.B</b>	Rotate right arithmetically
<b>Syntax</b>	RRA dst or RRA.W dst RRA.B dst

**Operation** MSB → MSB, MSB → MSB-1, ... LSB+1 → LSB, LSB → C

**Description** The destination operand is shifted right one position as shown in Figure 3-17. The MSB is shifted into the MSB, the MSB is shifted into the MSB-1, and the LSB+1 is shifted into the LSB.



**Figure 3-17. Destination Operand – Arithmetic Right Shift**

**Status Bits**

- N: Set if result is negative, reset if positive
- Z: Set if result is zero, reset otherwise
- C: Loaded from the LSB
- V: Reset

**Mode Bits** OSCOFF, CPUOFF, and GIE are not affected.

**Example** R5 is shifted right one position. The MSB retains the old value. It operates equal to an arithmetic division by 2.

```
RRA    R5    ; R5/2 -> R5
; The value in R5 is multiplied by 0.75 (0.5 + 0.25).
;
PUSH    R5    ; Hold R5 temporarily using stack
RRA     R5    ; R5 x 0.5 -> R5
ADD     @SP+,R5 ; R5 x 0.5 + R5 = 1.5 x R5 -> R5
RRA     R5    ; (1.5 x R5) x 0.5 = 0.75 x R5 -> R5
.....
```

**Example** The low byte of R5 is shifted right one position. The MSB retains the old value. It operates equal to an arithmetic division by 2.

```
RRA.B   R5    ; R5/2 -> R5: operation is on low byte only
; High byte of R5 is reset
PUSH.B  R5    ; R5 x 0.5 -> TOS
RRA.B   @SP    ; TOS x 0.5 = 0.5 x R5 x 0.5 = 0.25 x R5 -> TOS
ADD.B   @SP+,R5 ; R5 x 0.5 + R5 x 0.25 = 0.75 x R5 -> R5
.....
```

### 3.4.6.41 RRC

**RRC[.W]** Rotate right through carry


**RRC.B** Rotate right through carry

**Syntax**

```
RRC dst or RRC.W dst
RRC dst
```

**Operation**  $C \rightarrow \text{MSB} \rightarrow \text{MSB}-1 \dots \text{LSB}+1 \rightarrow \text{LSB} \rightarrow C$

**Description** The destination operand is shifted right one position as shown in Figure 3-18. The carry bit (C) is shifted into the MSB, the LSB is shifted into the carry bit (C).



**Figure 3-18. Destination Operand - Carry Right Shift**

**Status Bits**

N: Set if result is negative, reset if positive

Z: Set if result is zero, reset otherwise

C: Loaded from the LSB

V: Reset

**Mode Bits** OSCOFF, CPUOFF, and GIE are not affected.

**Example** R5 is shifted right one position. The MSB is loaded with 1.

```
SETC          ; Prepare carry for MSB
RRC    R5    ; R5/2 + 8000h -> R5
```

**Example** R5 is shifted right one position. The MSB is loaded with 1.

```
SETC          ; Prepare carry for MSB
RRC.B  R5    ; R5/2 + 80h -> R5; low byte of R5 is used
```

### 3.4.6.42 SBC

<b>*SBC[W]</b>	Subtract source and borrow/.NOT. carry from destination
<b>*SBC.B</b>	Subtract source and borrow/.NOT. carry from destination
<b>Syntax</b>	SBC dst or SBC.W dst SBC.B dst
<b>Operation</b>	dst + 0FFFFh + C → dst dst + 0FFh + C → dst
<b>Emulation</b>	SUBC #0, dst SUBC.B #0, dst
<b>Description</b>	The carry bit (C) is added to the destination operand minus one. The previous contents of the destination are lost.
<b>Status Bits</b>	N: Set if result is negative, reset if positive Z: Set if result is zero, reset otherwise C: Set if there is a carry from the MSB of the result, reset otherwise. Set to 1 if no borrow, reset if borrow. V: Set if an arithmetic overflow occurs, reset otherwise.
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	The 16-bit counter pointed to by R13 is subtracted from a 32-bit counter pointed to by R12.  SUB @R13, 0(R12) ; Subtract LSDs SBC 2(R12) ; Subtract carry from MSD
<b>Example</b>	The 8-bit counter pointed to by R13 is subtracted from a 16-bit counter pointed to by R12.  SUB.B @R13, 0(R12) ; Subtract LSDs SBC.B 1(R12) ; Subtract carry from MSD

---

**NOTE: Borrow Implementation**

The borrow is treated as a .NOT. carry:

Borrow	Carry bit
Yes	0
No	1

---

### 3.4.6.43 SETC

**\*SETC** Set carry bit

**Syntax** SETC

**Operation**  $1 \rightarrow C$

**Emulation** BIS #1,SR

**Description** The carry bit (C) is set.

**Status Bits** N: Not affected

Z: Not affected

C: Set

V: Not affected

**Mode Bits** OSCOFF, CPUOFF, and GIE are not affected.

**Example** Emulation of the decimal subtraction:

Subtract R5 from R6 decimally

Assume that R5 = 03987h and R6 = 04137h

```
DSUB   ADD    #06666h,R5    ; Move content R5 from 0-9 to 6-0Fh
                                ; R5 = 03987h + 06666h = 09FEDh
                                ; Invert this (result back to 0-9)
                                ; R5 = .NOT. R5 = 06012h
                                ; Prepare carry = 1
                                SETC
                                DADD   R5,R6    ; Emulate subtraction by addition of:
                                ; (010000h - R5 - 1)
                                ; R6 = R6 + R5 + 1
                                ; R6 = 0150h
```

---

**3.4.6.44 SETN**

---

<b>*SETN</b>	Set negative bit
<b>Syntax</b>	SETN
<b>Operation</b>	$1 \rightarrow N$
<b>Emulation</b>	BIS #4, SR
<b>Description</b>	The negative bit (N) is set.
<b>Status Bits</b>	N: Set Z: Not affected C: Not affected V: Not affected
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.

### 3.4.6.45 SETZ

---

<b>*SETZ</b>	Set zero bit
<b>Syntax</b>	SETZ
<b>Operation</b>	$1 \rightarrow Z$
<b>Emulation</b>	BIS #2, SR
<b>Description</b>	The zero bit (Z) is set.
<b>Status Bits</b>	N: Not affected Z: Set C: Not affected V: Not affected
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.

### 3.4.6.46 SUB

<b>SUB[W]</b>	Subtract source from destination
<b>SUB.B</b>	Subtract source from destination
<b>Syntax</b>	<pre>SUB src,dst or SUB.W src,dst SUB.B src,dst</pre>
<b>Operation</b>	$dst + \text{.NOT}.src + 1 \rightarrow dst$ or $[(dst - src) \rightarrow dst]$
<b>Description</b>	The source operand is subtracted from the destination operand by adding the source operand's 1s complement and the constant 1. The source operand is not affected. The previous contents of the destination are lost.
<b>Status Bits</b>	N: Set if result is negative, reset if positive Z: Set if result is zero, reset otherwise C: Set if there is a carry from the MSB of the result, reset otherwise. Set to 1 if no borrow, reset if borrow. V: Set if an arithmetic overflow occurs, otherwise reset
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	See example at the SBC instruction.
<b>Example</b>	See example at the SBC.B instruction.

---

**NOTE: Borrow Is Treated as a .NOT.**

The borrow is treated as a .NOT. carry:	Borrow	Carry bit
	Yes	0
	No	1

---

### 3.4.6.47 SUBC, SBB

<b>SUBC[.W], SBB[.W]</b>	Subtract source and borrow/.NOT. carry from destination
<b>SUBC.B, SBB.B</b>	Subtract source and borrow/.NOT. carry from destination
<b>Syntax</b>	<pre> SUBC      src,dst      or      SUBC.W  src,dst      or SBB       src,dst      or      SBB.W   src,dst SUBC.B    src,dst      or      SBB.B   src,dst </pre>
<b>Operation</b>	$dst + \text{.NOT.}src + C \rightarrow dst$ or $(dst - src - 1 + C \rightarrow dst)$
<b>Description</b>	The source operand is subtracted from the destination operand by adding the source operand's 1s complement and the carry bit (C). The source operand is not affected. The previous contents of the destination are lost.
<b>Status Bits</b>	N: Set if result is negative, reset if positive. Z: Set if result is zero, reset otherwise. C: Set if there is a carry from the MSB of the result, reset otherwise. Set to 1 if no borrow, reset if borrow. V: Set if an arithmetic overflow occurs, reset otherwise.
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	Two floating point mantissas (24 bits) are subtracted. LSBs are in R13 and R10, MSBs are in R12 and R9. <pre> SUB.W     R13,R10      ; 16-bit part, LSBs SUBC.B    R12,R9       ; 8-bit part, MSBs </pre>
<b>Example</b>	The 16-bit counter pointed to by R13 is subtracted from a 16-bit counter in R10 and R11(MSD). <pre> SUB.B     @R13+,R10    ; Subtract LSDs without carry SUBC.B    @R13,R11     ; Subtract MSDs with carry ... </pre>

---

**NOTE: Borrow Implementation**

The borrow is treated as a .NOT. carry:	Borrow	Carry bit
	Yes	0
	No	1

---

### 3.4.6.48 SWPB

**SWPB**

Swap bytes

**Syntax**

SWPB dst

**Operation**


Bits 15 to 8  $\leftrightarrow$  bits 7 to 0

**Description**

The destination operand high and low bytes are exchanged as shown in [Figure 3-19](#).

**Mode Bits**

OSCOFF, CPUOFF, and GIE are not affected.



**Figure 3-19. Destination Operand - Byte Swap**

**Example**

```
MOV    #040BFh,R7    ; 0100000010111111 -> R7
SWPB   R7              ; 1011111101000000 in R7
```

**Example**

The value in R5 is multiplied by 256. The result is stored in R5,R4.

```
SWPB   R5              ;
MOV     R5,R4          ; Copy the swapped value to R4
BIC     #0FF00h,R5     ; Correct the result
BIC     #00FFh,R4      ; Correct the result
```

### 3.4.6.49 SXT

**SXT** Extend Sign

**Syntax** SXT dst


**Operation** Bit 7 → Bit 8 ..... Bit 15

**Description** The sign of the low byte is extended into the high byte as shown in Figure 3-20.

**Status Bits**

- N: Set if result is negative, reset if positive
- Z: Set if result is zero, reset otherwise
- C: Set if result is not zero, reset otherwise (.NOT. Zero)
- V: Reset

**Mode Bits** OSCOFF, CPUOFF, and GIE are not affected.



**Figure 3-20. Destination Operand - Sign Extension**

**Example** R7 is loaded with the P1IN value. The operation of the sign-extend instruction expands bit 8 to bit 15 with the value of bit 7.

R7 is then added to R6.

```
MOV.B    &P1IN,R7    ; P1IN = 080h:    .... 1000 0000
SXT      R7           ; R7 = 0FF80h:    1111 1111 1000 0000
```

### 3.4.6.50 TST

<b>*TST[.W]</b>	Test destination
<b>*TST.B</b>	Test destination
<b>Syntax</b>	TST dst or TST.W dst TST.B dst
<b>Operation</b>	dst + 0FFFFh + 1 dst + 0FFh + 1
<b>Emulation</b>	CMP #0, dst CMP.B #0, dst
<b>Description</b>	The destination operand is compared with zero. The status bits are set according to the result. The destination is not affected.
<b>Status Bits</b>	N: Set if destination is negative, reset if positive Z: Set if destination contains zero, reset otherwise C: Set V: Reset
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	R7 is tested. If it is negative, continue at R7NEG; if it is positive but not zero, continue at R7POS.  <pre> TST      R7          ; Test R7 JN       R7NEG       ; R7 is negative JZ       R7ZERO      ; R7 is zero R7POS    .....      ; R7 is positive but not zero R7NEG    .....      ; R7 is negative R7ZERO   .....      ; R7 is zero </pre>
<b>Example</b>	The low byte of R7 is tested. If it is negative, continue at R7NEG; if it is positive but not zero, continue at R7POS.  <pre> TST.B    R7          ; Test low byte of R7 JN       R7NEG       ; Low byte of R7 is negative JZ       R7ZERO      ; Low byte of R7 is zero R7POS    .....      ; Low byte of R7 is positive but not zero R7NEG    .....      ; Low byte of R7 is negative R7ZERO   .....      ; Low byte of R7 is zero </pre>

### 3.4.6.51 XOR

<b>XOR[.W]</b>	Exclusive OR of source with destination
<b>XOR.B</b>	Exclusive OR of source with destination
<b>Syntax</b>	<pre>XOR src,dst or XOR.W src,dst XOR.B src,dst</pre>
<b>Operation</b>	src .XOR. dst → dst
<b>Description</b>	The source and destination operands are exclusive ORed. The result is placed into the destination. The source operand is not affected.
<b>Status Bits</b>	<p>N: Set if result MSB is set, reset if not set</p> <p>Z: Set if result is zero, reset otherwise</p> <p>C: Set if result is not zero, reset otherwise ( = .NOT. Zero)</p> <p>V: Set if both operands are negative</p>
<b>Mode Bits</b>	OSCOFF, CPUOFF, and GIE are not affected.
<b>Example</b>	<p>The bits set in R6 toggle the bits in the RAM word TONI.</p> <pre>XOR      R6,TONI ; Toggle bits of word TONI on the bits set in R6</pre>
<b>Example</b>	<p>The bits set in R6 toggle the bits in the RAM byte TONI.</p> <pre>XOR.B    R6,TONI ; Toggle bits of byte TONI on the bits set in                   ; low byte of R6</pre>
<b>Example</b>	<p>Reset to 0 those bits in low byte of R7 that are different from bits in RAM byte EDE.</p> <pre>XOR.B    EDE,R7  ; Set different bit to "1s" INV.B    R7       ; Invert Lowbyte, Highbyte is 0h</pre>